
enulisRANCANG BANGUN SISTEM DETEKSI ASAP ROKOK DI DALAM 

RUANGAN MENGGUNAKAN ANTENA MIKROSTRIP PATCH 

RECTANGULAR ARRAY 2,3 GHz DENGAN BOT TELEGRAM” 

 

ABSTRAK 

Meningkatnya paparan asap rokok di dalam ruangan tertutup dapat menimbulkan gangguan 

kesehatan bagi penghuni, terutama perokok pasif. Oleh karena itu, diperlukan suatu sistem 

yang mampu mendeteksi keberadaan asap rokok secara cepat dan memberikan notifikasi 

secara real-time. Penelitian ini bertujuan untuk merancang dan merealisasikan sistem 

deteksi asap rokok berbasis Internet of Things (IoT) menggunakan sensor MQ-2, ESP32 

DevKit V1, LCD I2C 16×2, buzzer, dan Bot Telegram. Sistem komunikasi didukung oleh 

antena mikrostrip patch rectangular array 1×2 U-Slot yang dirancang pada frekuensi 2,3 

GHz menggunakan substrat FR-4 dan diuji menggunakan Vector Network Analyzer 

(VNA), sedangkan pengujian pola radiasi antena dilakukan di dalam anechoic chamber 

pada Laboratorium Telekomunikasi Radio dan Gelombang Mikro (TRGM), Sekolah 

Teknik Elektro dan Informatika, Institut Teknologi Bandung (STEI-ITB), Bandung. Hasil 

pengukuran menunjukkan bahwa antena bekerja pada frekuensi resonansi 2,3 GHz dengan 

nilai return loss sebesar -27,62 dB, VSWR sebesar 1,086, bandwidth sebesar 82,46 MHz, 

dan gain sebesar 5,212 dBi. Hasil pengujian sistem menunjukkan bahwa sensor MQ-2 

membaca nilai smoke sebesar 62 (melebihi nilai threshold sebesar 50) saat mendeteksi asap 

rokok, sehingga ESP32 mengaktifkan buzzer, menampilkan status "BAHAYA" beserta 

nilai smoke dan threshold pada LCD I2C, serta mengirimkan notifikasi "ALARM ASAP 

ROKOK" melalui Bot Telegram secara real-time. Sebaliknya, pada kondisi udara normal 

dengan nilai smoke sebesar 16 (di bawah threshold), sistem menampilkan status 

"NORMAL" pada LCD tanpa mengaktifkan buzzer maupun mengirimkan notifikasi 

bahaya. Berdasarkan hasil pengujian yang telah dilakukan, sistem deteksi asap rokok yang 

dirancang mampu bekerja dengan baik dan dapat digunakan sebagai solusi pemantauan 

kualitas udara dalam ruangan secara otomatis. 

 

Kata kunci: Antena Mikrostrip, Patch Rectangular Array, 2,3 GHz, Deteksi Asap Rokok, 

MQ-2, ESP32, Bot Telegram, Internet of Things (IoT). 

“DESIGN AND IMPLEMENTATION OF INDOOR CIGARETTE SMOKE 

DETECTION SYSTEM USING 2,3 GHz RECTANGULAR PATCH 

MICROSTRIP ANTENNA ARRAY WITH TELEGRAM BOT” 

 

ABSTRACT 

The increasing exposure to cigarette smoke in enclosed spaces can cause health problems 

for occupants, particularly passive smokers. Therefore, a system capable of rapidly 

detecting the presence of cigarette smoke and providing real-time notifications is required. 

This study aims to design and implement an Internet of Things (IoT)-based cigarette smoke 

detection system using an MQ-2 sensor, ESP32 DevKit V1, 16×2 I2C LCD, buzzer, and 

Telegram Bot. The communication system is supported by a 2.3 GHz 1×2 U-Slot 

rectangular patch array microstrip antenna designed on an FR-4 substrate and 

characterized using a Vector Network Analyzer (VNA), while the antenna radiation pattern 

testing was carried out inside an anechoic chamber at the Radio and Microwave 

Telecommunications Laboratory (TRGM), School of Electrical Engineering and 

Informatics, Institut Teknologi Bandung (STEI-ITB), Bandung. The measurement results 

show that the antenna operates at a resonant frequency of 2.3 GHz, with a return loss of 

−27.62 dB, a VSWR of 1.086, a bandwidth of 82.46 MHz, and a gain of 5.212 dBi. The 

system testing results show that the MQ-2 sensor recorded a smoke value of 62 (exceeding 

the threshold value of 50) upon detecting cigarette smoke, causing the ESP32 to activate 

the buzzer, display the "DANGER" status along with the smoke and threshold values on the 

I2C LCD, and send an "SMOKE ALARM" notification via the Telegram Bot in real time. 

Conversely, under normal air conditions with a smoke value of 16 (below the threshold), 

the system displayed a "NORMAL" status on the LCD without activating the buzzer or 

sending a danger notification. Based on the conducted tests, the proposed cigarette smoke 

detection system performs effectively and can be utilized as an automated indoor air quality 

monitoring solution. 

 

Keywords: Microstrip Antenna, Rectangular Patch Array, 2,3 GHz, Cigarette Smoke 

Detection, MQ-2, ESP32, Telegram Bot, Internet of Things (IoT). 

BAB I 

PENDAHULUAN 

1.1 Latar Belakang 

Asap rokok merupakan salah satu polutan udara dalam ruangan yang paling 

berbahaya bagi kesehatan manusia. Menurut World Health Organization (WHO), 

paparan asap rokok (second-hand smoke) di dalam ruangan tertutup dapat 

meningkatkan risiko penyakit jantung koroner sebesar 25–30% dan kanker paru-

paru sebesar 20–30% pada individu yang tidak merokok, serta menyebabkan iritasi 

mata, hidung, tenggorokan, dan paru-paru dalam jangka pendek (World Health 

Organization, n.d.). Banyaknya fasilitas umum, perkantoran, dan hunian yang 

masih rentan terhadap pelanggaran kawasan bebas rokok menjadikan sistem deteksi 

asap rokok otomatis sebagai kebutuhan yang semakin mendesak. 

Seiring berkembangnya teknologi Internet of Things (IoT), kini sistem 

pemantauan lingkungan dapat dibuat lebih efisien dan mudah diakses dari jarak 

jauh menggunakan mikrokontroler ESP32 dan notifikasi Bot Telegram. Namun, 

antena bawaan pada modul ESP32 memiliki keterbatasan dalam hal gain dan 

jangkauan sinyal WiFi pada lingkungan indoor yang memiliki banyak penghalang 

sehingga kualitas komunikasi dapat menurun (Espressif Systems, 2023). Untuk 

mengatasi hal tersebut, digunakan antena mikrostrip patch rectangular array 1x2 

elemen pada frekuensi 2,3 GHz yang memiliki profil rendah, bobot ringan, ukuran 

kompak, dan kemampuan meningkatkan gain sinyal WiFi. 

Sistem ini mengintegrasikan sensor gas MQ-2, ESP32 DevKit V1, LCD I2C 

16x2, buzzer, dan notifikasi Bot Telegram melalui jaringan WiFi. Diharapkan 

sistem ini dapat membantu penegakan kawasan bebas rokok dan melindungi 

kesehatan penghuni ruangan secara efektif dan efisien. 

1.2 Perumusan Masalah 

Dari latar belakang di atas dapat ditarik suatu permasalahan yaitu: 

1. Bagaimana cara merancang antena mikrostrip rectangular patch array 1x2 

pada frekuensi 2,3 GHz menggunakan CST Studio Suite 2019? 

2. Bagaimana karakteristik antena mikrostrip rectangular patch array hasil 

pengukuran VNA, meliputi return loss (S11), VSWR, impedansi input, 

gain dan hasil pola radiasi? 

3. Bagaimana kinerja sistem deteksi asap rokok berbasis IoT dalam 

mendeteksi keberadaan asap rokok dan mengirimkan notifikasi secara 

real-time melalui Bot Telegram? 

4. Bagaimana pengaruh penggunaan antena mikrostrip terhadap kualitas 

layanan (Quality of Service) QoS jaringan WiFi yang digunakan pada 

sistem deteksi asap rokok berbasis IoT? 

1.3 Tujuan 

Tujuan yang ingin dicapai dari tugas akhir ini adalah sebagai berikut : 

1. Dapat merancang antena mikrostrip rectangular patch array 1x2 pada 

frekuensi 2,3 GHz menggunakan perangkat lunak CST Studio Suite 2019.  

2. Dapat mengetahui karakteristik antena mikrostrip rectangular patch array 

1x2 meliputi return loss, VSWR, impedansi input, gain, dan pola radiasi 

berdasarkan hasil pengukuran VNA. 

3. Dapat menganalisis kinerja sistem deteksi asap rokok berbasis IoT dalam 

mendeteksi keberadaan asap rokok dan mengirimkan notifikasi secara 

real-time melalui Bot Telegram. 

4. Dapat menganalisis pengaruh penggunaan antena mikrostrip terhadap 

kualitas layanan (Quality of Service/QoS) jaringan WiFi yang digunakan 

pada sistem deteksi asap rokok berbasis IoT. 

1.4 Luaran 

Luaran dari tugas akhir ini adalah sebagai berikut : 

1. Antena mikrostrip rectangular patch array 1x2 dengan frekuensi kerja 2,3 

GHz. 

2. Prototipe sistem deteksi asap rokok berbasis Internet of Things (IoT). 

3. Laporan Tugas Akhir. 

4. Artikel ilmiah pada Jurnal Nasional Terakreditasi 

 



 

3 

Politeknik Negeri Jakarta 

BAB V 

PENUTUP 

1.1 Kesimpulan 

1. Antena mikrostrip patch array 1×2 U-Slot frekuensi 2,3 GHz berhasil 

dirancang, direalisasikan, dan dipasang pada Router WiFi Orbit sebagai 

antena eksternal untuk mendukung komunikasi sistem deteksi asap rokok 

berbasis IoT.  

2. Berdasarkan hasil pengukuran menggunakan Vector Network Analyzer 

(VNA), karakteristik antena mikrostrip rectangular patch array 1×2 

pada frekuensi 2,3 GHz telah diketahui, meliputi nilai return loss sebesar 

−27,62 dB, VSWR sebesar 1,086, impedansi input sebesar 50,20 Ω, gain 

sebesar 5,212 dBi, serta pola radiasi directional. Hasil tersebut 

menunjukkan bahwa antena telah memenuhi spesifikasi perancangan. 

3. Sistem deteksi asap rokok bekerja dengan baik. Sensor MQ-2 mampu 

mendeteksi asap, sedangkan ESP32 berhasil mengaktifkan buzzer, 

menampilkan status pada LCD I2C, dan mengirimkan notifikasi ke Bot 

Telegram secara real-time. 

4. Berdasarkan hasil pengujian QoS, diperoleh nilai rata-rata throughput 

25,34 kbps, delay 86,78 ms, jitter 114,81 ms, dan packet loss 0%, 

sehingga komunikasi data antara ESP32 dan Bot Telegram berjalan stabil 

dan real-time. 

1.2 Saran 

1. Pada penelitian selanjutnya, antena dapat dikembangkan menggunakan 

material substrat dengan rugi-rugi (loss tangent) yang lebih rendah agar 

nilai gain dan efisiensi antena dapat ditingkatkan. 

2. Sistem dapat dikembangkan dengan menambahkan lebih banyak sensor 

atau menggunakan sensor gas yang memiliki tingkat sensitivitas dan 

selektivitas lebih baik sehingga pendeteksian asap menjadi lebih akurat. 

3. Bot Telegram dapat dikembangkan dengan menambahkan fitur riwayat 

notifikasi agar informasi yang diterima pengguna lebih lengkap. 

4. Pengujian Quality of Service (QoS) dapat dilakukan pada lebih banyak 

variasi kondisi jaringan, jarak, dan jumlah pengguna untuk memperoleh 

hasil analisis komunikasi data yang lebih menyeluruh. 

LAMPIRAN 

Lampiran 1 Source Code ESP32 

/**************************************************** 
 * SMOKE DETECTOR ESP32 - FINAL FIX INSTAN ON/OFF 
 * Standar Kualitas Udara: > 50 PPM (Tidak Sehat/Bahaya) 
 * Hardware PCB (image_1df042.png) Matched. 
 * UPDATE: HYPER-SENSITIVE MODE 
 ****************************************************/ 
 
#include <WiFi.h> 
#include <WiFiClientSecure.h>  
#include <UniversalTelegramBot.h> 
#include <Wire.h> 
#include <LiquidCrystal_I2C.h> 
 
//============================== 
// WIFI 
//============================== 
const char* ssid     = "rell"; 
const char* password = "darelazzah"; 
 
//============================== 
// TELEGRAM CREDENTIALS 
//============================== 
#define BOT_TOKEN "8450190981:AAFACozZQBIaT89sxHByAa2nGUTXSJZOcdk" 
String CHAT_ID = "8398821948"; 
 
WiFiClientSecure client; 
UniversalTelegramBot bot(BOT_TOKEN, client); 
 
//============================== 
// LCD 
//============================== 
LiquidCrystal_I2C lcd(0x27, 16, 2); 
 
//============================== 
// PIN (Sesuai Layout PCB) 
//============================== 
#define MQ2_PIN       34 
#define TRIMPOT_PIN   35 
#define BUZZER_PIN    4 
#define LED_GREEN     26 
#define LED_RED       27 
 
//============================== 
// VARIABLE 
//============================== 
int smokeValue   = 0;    
int smokePpm     = 0;    
int rssi         = 0;    
 
bool alarmActive   = false; 
bool telegramSent  = false; 
 
unsigned long lastLCD  = 0; 
unsigned long lastRead = 0; 
 
//============================== 
// KALIBRASI PPM (DIUBAH BIAR LEBIH SENSITIF) 
//============================== 
// Semakin kecil angka LEVEL_ACTIVE_SMOKE, semakin cepat PPM meroket 

const float LEVEL_ACTIVE_SMOKE = 15.0;   // Asalnya 46.0, diturunin drastis jadi 15.0 
const float PPM_ACTIVE_SMOKE   = 40.0;    
const float PPM_GAIN = PPM_ACTIVE_SMOKE / LEVEL_ACTIVE_SMOKE;  
 
//============================== 
// BASELINE 
//============================== 
int baselineSmoke = 0;       
int smokeLevel    = 0;       
int sensitivity   = 51;     
int alarmTarget   = 0;       
 
//============================== 
// FILTER 
//============================== 
const byte SAMPLE = 10; 
int sampleBuffer[SAMPLE]; 
 
const byte TRIM_SAMPLE = 15; 
int trimBuffer[TRIM_SAMPLE]; 
int trimIndex = 0; 
long trimSum = 0; 
bool trimBufferFull = false; 
 
//============================== 
// MANUAL TEST MODE 
//============================== 
bool manualTestActive = false; 
unsigned long manualTestStart = 0; 
const unsigned long MANUAL_TEST_DURATION = 3000;  
 
//============================== 
// TELEGRAM TIMEOUT / STATUS 
//============================== 
const uint16_t TELEGRAM_HANDSHAKE_TIMEOUT_SEC = 12;   
const uint16_t TELEGRAM_CONNECT_TIMEOUT_MS    = 8000; 
bool telegramReachable = false;  
 
//============================== 
// PROTOTYPE 
//============================== 
void connectWiFi(); 
void warmupSensor(); 
void calibrateMQ2(); 
void readSensor(); 
void updateLCD(); 
void normalMode(); 
void alarmMode(); 
void checkAlarm(); 
void buzzerSelfTest(); 
void handleSerialTest(); 
void sendTelegramAlarm(); 
void sendTelegramNormal(); 
void reconnectWiFi(); 
int filteredMQ2(); 
int filteredTrimpot(); 
bool testTelegramReachability(); 
void recalibrateNow(); 
 
//================================================== 
// SETUP 
//================================================== 
void setup() 
{ 
  Serial.begin(115200); 
  Serial.println("Ketik 'T' lalu Enter untuk TEST manual alarm."); 
  Serial.println("Ketik 'C' lalu Enter untuk KALIBRASI ULANG baseline."); 

  pinMode(LED_GREEN, OUTPUT); 
  pinMode(LED_RED, OUTPUT); 
  pinMode(BUZZER_PIN, OUTPUT); 
 
  digitalWrite(LED_GREEN, LOW); 
  digitalWrite(LED_RED, LOW); 
  digitalWrite(BUZZER_PIN, LOW); 
 
  Wire.begin(21, 22); 
  lcd.init(); 
  lcd.backlight(); 
 
  lcd.clear(); 
  lcd.setCursor(0, 0); 
  lcd.print("Smoke Detector"); 
  lcd.setCursor(0, 1); 
  lcd.print("Starting..."); 
  delay(1500); 
 
  analogReadResolution(12);         
  analogSetAttenuation(ADC_11db); 
 
  for (int i = 0; i < TRIM_SAMPLE; i++) 
  { 
    trimBuffer[i] = analogRead(TRIMPOT_PIN); 
    trimSum += trimBuffer[i]; 
    delay(10); 
  } 
  trimBufferFull = true; 
 
  connectWiFi(); 
   
  client.setInsecure(); 
  client.setHandshakeTimeout(TELEGRAM_HANDSHAKE_TIMEOUT_SEC);  
  client.setTimeout(TELEGRAM_CONNECT_TIMEOUT_MS / 1000);       
 
  warmupSensor(); 
  calibrateMQ2(); 
 
  buzzerSelfTest(); 
  lcd.clear(); 
 
  if (WiFi.status() == WL_CONNECTED) { 
 
    lcd.setCursor(0, 0); 
    lcd.print("Cek Telegram... "); 
    telegramReachable = testTelegramReachability(); 
    lcd.setCursor(0, 1); 
    lcd.print(telegramReachable ? "Server OK       " : "Server GAGAL    "); 
    delay(1500); 
 
    if (telegramReachable) { 
      Serial.println(">>> Mencoba kirim pesan Booting ke Telegram..."); 
      client.setInsecure();  
 
      String bootMsg = "\xF0\x9F\xA4\x96 Alat Smoke Detector Berhasil Dinyalakan!\n"; 
      bootMsg += "Sinyal WiFi: " + String(WiFi.RSSI()) + " dBm"; 
      bool sent = bot.sendMessage(CHAT_ID, bootMsg, ""); 
 
      if (sent) { 
        Serial.println(">>> Pesan Booting BERHASIL terkirim!"); 
      } else { 
        Serial.println(">>> Pesan Booting GAGAL terkirim!"); 

      } 
    } else { 
      Serial.println(">>> DIAGNOSTIK: ESP32 TIDAK BISA MENJANGKAU api.telegram.org."); 
    } 
  } 
} 
 
//================================================== 
// DIAGNOSTIK: TES KONEKSI KE SERVER TELEGRAM 
//================================================== 
bool testTelegramReachability() 
{ 
  WiFiClientSecure testClient; 
  testClient.setInsecure(); 
  testClient.setHandshakeTimeout(TELEGRAM_HANDSHAKE_TIMEOUT_SEC); 
 
  unsigned long t0 = millis(); 
  Serial.print(">>> Diagnostik: menghubungi api.telegram.org ... "); 
 
  bool ok = testClient.connect("api.telegram.org", 443); 
  unsigned long elapsed = millis() - t0; 
 
  if (ok) { 
    Serial.print("BERHASIL ("); 
    Serial.print(elapsed); 
    Serial.println(" ms)"); 
    testClient.stop(); 
    return true; 
  } else { 
    Serial.print("GAGAL setelah "); 
    Serial.print(elapsed); 
    Serial.println(" ms"); 
    testClient.stop(); 
    return false; 
  } 
} 
 
//================================================== 
// BUZZER SELF TEST 
//================================================== 
void buzzerSelfTest() 
{ 
  lcd.clear(); 
  lcd.setCursor(0, 0); 
  lcd.print("Test Buzzer..."); 
 
  for (int i = 0; i < 2; i++) 
  { 
    digitalWrite(BUZZER_PIN, HIGH); 
    delay(150); 
    digitalWrite(BUZZER_PIN, LOW); 
    delay(150); 
  } 
  delay(500); 
} 
 
//================================================== 
// CONNECT WIFI  
//================================================== 
void connectWiFi() 
{ 
  WiFi.mode(WIFI_STA); 
  WiFi.begin(ssid, password); 
 
  lcd.clear(); 
  lcd.setCursor(0, 0); 
  lcd.print("Connecting WiFi"); 

  int dot = 0; 
  unsigned long start = millis(); 
 
  while (WiFi.status() != WL_CONNECTED && millis() - start < 15000) 
  { 
    delay(500); 
    lcd.setCursor(dot, 1); 
    lcd.print("."); 
    dot++; 
    if (dot > 15) 
    { 
      lcd.setCursor(0, 1); 
      lcd.print("                "); 
      dot = 0; 
    } 
  } 
 
  lcd.clear(); 
  lcd.setCursor(0, 0); 
 
  if (WiFi.status() == WL_CONNECTED) 
  { 
    lcd.print("WiFi Connected"); 
    lcd.setCursor(0, 1); 
    lcd.print(WiFi.localIP()); 
  } 
  else 
  { 
    lcd.print("WiFi Timeout"); 
    lcd.setCursor(0, 1); 
    lcd.print("Offline Mode"); 
  } 
  delay(2000); 
} 
 
//================================================== 
// KALIBRASI MQ2 
//================================================== 
void calibrateMQ2() 
{ 
  lcd.clear(); 
  lcd.setCursor(0, 0); 
  lcd.print("Kalibrasi..."); 
 
  long total = 0; 
 
  for (int i = 0; i < 100; i++) 
  { 
    total += filteredMQ2(); 
    lcd.setCursor(0, 1); 
    lcd.print(i + 1); 
    lcd.print("%   "); 
    delay(50); 
  } 
 
  baselineSmoke = total / 100; 
 
  lcd.clear(); 
  lcd.setCursor(0, 0); 
  lcd.print("Baseline:"); 
  lcd.setCursor(0, 1); 
  lcd.print(baselineSmoke); 
  delay(2000); 
} 

//================================================== 
// WARMUP MQ2 (25 detik) 
//================================================== 
void warmupSensor() 
{ 
  const int WARMUP_SECONDS = 25; 
  for (int i = WARMUP_SECONDS; i >= 0; i--) 
  { 
    lcd.clear(); 
    lcd.setCursor(0, 0); 
    lcd.print("MQ2 Warming Up"); 
    lcd.setCursor(0, 1); 
    lcd.print(i); 
    lcd.print(" sec"); 
    delay(1000); 
  } 
  lcd.clear(); 
} 
 
//================================================== 
// MOVING AVERAGE FILTER 
//================================================== 
int filteredMQ2() 
{ 
  long total = 0; 
  for (int i = 0; i < SAMPLE; i++) 
  { 
    total += analogRead(MQ2_PIN); 
    delay(5); 
  } 
  return total / SAMPLE; 
} 
 
int filteredTrimpot() 
{ 
  int raw = analogRead(TRIMPOT_PIN); 
  trimSum -= trimBuffer[trimIndex]; 
  trimBuffer[trimIndex] = raw; 
  trimSum += raw; 
  trimIndex = (trimIndex + 1) % TRIM_SAMPLE; 
  return trimSum / TRIM_SAMPLE; 
} 
 
//================================================== 
// READ SENSOR  
//================================================== 
void readSensor() 
{ 
  smokeValue = filteredMQ2(); 
 
  // AUTO-TRACKING BASELINE  
  if (smokeValue < baselineSmoke - 2) { 
    baselineSmoke = smokeValue;  
  } 
 
  smokeLevel = smokeValue - baselineSmoke; 
  if (smokeLevel < 0) smokeLevel = 0; 
 
  // DEADBAND FILTER (DIUBAH BIAR LEBIH SENSITIF) 
  // Sekarang fluktuasi kecil (>= 2) langsung dibaca jadi PPM 
  if (smokeLevel < 2) {  
    smokeLevel = 0;  
  } 
 
  smokePpm = (int)round(smokeLevel * PPM_GAIN); 
 
  int trimFiltered = filteredTrimpot(); 

  // RANGE TRIMPOT DIGANTI: Sesuai standar 30 PPM sampai 100 PPM 
  // Putar Trimpot sampai layar menunjukkan 52 PPM (Batas Tidak Sehat) 
  sensitivity = map(trimFiltered, 0, 4095, 30, 100); 
 
  alarmTarget = sensitivity; 
 
  if (WiFi.status() == WL_CONNECTED) 
    rssi = WiFi.RSSI(); 
  else 
    rssi = 0; 
 
  float voltage = smokeValue * 3.3 / 4095.0; 
 
  Serial.print("ADC=");        Serial.print(smokeValue); 
  Serial.print("  Volt=");     Serial.print(voltage, 2); 
  Serial.print("  RawLvl=");   Serial.print(smokeLevel); 
  Serial.print("  PPM=");      Serial.print(smokePpm); 
  Serial.print("  Baseline="); Serial.print(baselineSmoke); 
  Serial.print("  ThresholdPPM="); Serial.print(sensitivity); 
  Serial.print("  RSSI=");     Serial.println(rssi); 
} 
 
//================================================== 
// LCD  
//================================================== 
void updateLCD() 
{ 
  lcd.setCursor(0, 0); 
 
  if (manualTestActive) 
  { 
    lcd.print("*** TEST MODE *** "); 
  } 
  else 
  { 
    String line1 = "Smoke:" + String(smokePpm) + "/" + String(sensitivity) + "PPM"; 
     
    while(line1.length() < 16) { 
      line1 += " "; 
    } 
    lcd.print(line1.substring(0, 16)); 
  } 
 
  String status = alarmActive ? "BAHAYA  " : "AMAN    "; 
   
  String line2 = status + "Sig:" + String(rssi); 
  while (line2.length() < 16) line2 += " "; 
 
  lcd.setCursor(0, 1); 
  lcd.print(line2.substring(0, 16)); 
} 
 
//================================================== 
// KENDALI ALARM 
//================================================== 
void normalMode() 
{ 
  digitalWrite(LED_GREEN, HIGH); 
  digitalWrite(LED_RED, LOW); 
  digitalWrite(BUZZER_PIN, LOW); 
} 
 
void alarmMode() 
{ 

  digitalWrite(LED_GREEN, LOW); 
  digitalWrite(LED_RED, HIGH); 
  digitalWrite(BUZZER_PIN, HIGH); 
} 
 
//================================================== 
// LOGIKA ALARM INSTAN 100% (TANPA DELAY / HYSTERESIS) 
//================================================== 
void checkAlarm() 
{ 
  if (manualTestActive) return; 
 
  // Jika PPM Naik menyentuh atau melewati Threshold (Misal 52) 
  if (smokePpm >= sensitivity)  
  { 
    if (!alarmActive)  
    { 
      alarmActive = true; 
      alarmMode();    // Buzzer INSTAN NYALA 
      updateLCD(); 
 
      // Kirim Telegram 
      if (!telegramSent) { 
        telegramSent = true; 
        sendTelegramAlarm();  
      } 
    }  
    else  
    { 
      alarmMode();    // Pastikan Buzzer terus nyala selama asap masih ada 
    } 
  } 
  // Jika PPM turun (Walaupun cuma turun 1 angka dari batas) 
  else  
  { 
    if (alarmActive)  
    { 
      alarmActive = false; 
      normalMode();   // Buzzer INSTAN MATI 
      updateLCD(); 
 
      // Kirim Telegram Normal 
      if (telegramSent) { 
        telegramSent = false; 
        sendTelegramNormal();  
      } 
    }  
    else  
    { 
      normalMode();   // Pastikan Buzzer mati sepenuhnya saat aman 
    } 
  } 
} 
 
//================================================== 
// MANUAL TEST VIA SERIAL MONITOR 
//================================================== 
void handleSerialTest() 
{ 
  if (Serial.available()) 
  { 
    char c = Serial.read(); 
    if ((c == 'T' || c == 't') && !manualTestActive) 
    { 
      manualTestActive = true; 
      manualTestStart = millis(); 
 
      Serial.println(">>> MANUAL TEST DIMULAI: LED + Buzzer + Telegram <<<"); 

      alarmActive = true; 
      alarmMode(); 
      updateLCD(); 
       
      sendTelegramAlarm();  
    } 
    else if (c == 'C' || c == 'c') 
    { 
      recalibrateNow(); 
    } 
  } 
 
  if (manualTestActive && (millis() - manualTestStart >= MANUAL_TEST_DURATION)) 
  { 
    manualTestActive = false; 
    alarmActive = false; 
    normalMode(); 
    updateLCD(); 
 
    Serial.println(">>> MANUAL TEST SELESAI <<<"); 
    sendTelegramNormal();  
  } 
} 
 
//================================================== 
// KALIBRASI ULANG BASELINE TANPA RESTART 
//================================================== 
void recalibrateNow() 
{ 
  Serial.println(">>> KALIBRASI ULANG DIMULAI - pastikan udara bersih..."); 
 
  lcd.clear(); 
  lcd.setCursor(0, 0); 
  lcd.print("Kalibrasi Ulang"); 
 
  long total = 0; 
  for (int i = 0; i < 100; i++) 
  { 
    total += filteredMQ2(); 
    lcd.setCursor(0, 1); 
    lcd.print(i + 1); 
    lcd.print("%   "); 
    delay(50); 
  } 
 
  baselineSmoke = total / 100; 
 
  Serial.print(">>> Baseline baru = "); 
  Serial.println(baselineSmoke); 
 
  lcd.clear(); 
} 
 
//================================================== 
// TELEGRAM ALARM  
//================================================== 
void sendTelegramAlarm() 
{ 
  if (WiFi.status() != WL_CONNECTED) return; 
  if (!telegramReachable) { 
    Serial.println(">>> Lewati kirim Telegram: server tidak terjangkau."); 
    return; 
  } 

  Serial.println(">>> Sedang mengirim pesan ALARM ke Telegram..."); 
 
  client.setInsecure(); 
 
  String message = "\xF0\x9F\x9A\xA8 ALARM ASAP TERDETEKSI\n"; 
  message += "====================\n"; 
  message += "Kadar Smoke : " + String(smokePpm) + " PPM\n"; 
  message += "Threshold   : " + String(alarmTarget) + " PPM\n"; 
  message += "Kekuatan Sinyal: " + String(rssi) + " dBm\n"; 
  message += "Status      : BAHAYA"; 
 
  bool sent = bot.sendMessage(CHAT_ID, message, ""); 
   
  if(sent) { 
    Serial.println(">>> Telegram ALARM BERHASIL terkirim!"); 
  } else { 
    Serial.println(">>> Telegram ALARM GAGAL terkirim!"); 
  } 
} 
 
//================================================== 
// TELEGRAM NORMAL  
//================================================== 
void sendTelegramNormal() 
{ 
  if (WiFi.status() != WL_CONNECTED) return; 
  if (!telegramReachable) { 
    Serial.println(">>> Lewati kirim Telegram: server tidak terjangkau."); 
    return; 
  } 
 
  Serial.println(">>> Sedang mengirim pesan NORMAL ke Telegram..."); 
 
  client.setInsecure(); 
 
  String message = "\xE2\x9C\x85 KONDISI KEMBALI NORMAL\n"; 
  message += "====================\n"; 
  message += "Kadar Smoke : " + String(smokePpm) + " PPM\n"; 
  message += "Threshold   : " + String(alarmTarget) + " PPM\n"; 
  message += "Kekuatan Sinyal: " + String(rssi) + " dBm\n"; 
  message += "Status      : AMAN"; 
 
  bool sent = bot.sendMessage(CHAT_ID, message, ""); 
   
  if(sent) { 
    Serial.println(">>> Telegram NORMAL BERHASIL terkirim!"); 
  } else { 
    Serial.println(">>> Telegram NORMAL GAGAL terkirim!"); 
  } 
} 
 
//================================================== 
// AUTO RECONNECT WIFI 
//================================================== 
void reconnectWiFi() 
{ 
  if (WiFi.status() == WL_CONNECTED) return; 
 
  static unsigned long lastAttempt = 0; 
  if (millis() - lastAttempt < 15000) return; 
 
  lastAttempt = millis(); 
  WiFi.disconnect(true); 
  delay(500); 
  WiFi.begin(ssid, password); 

} 
 
//================================================== 
// LOOP 
//================================================== 
void loop() 
{ 
  reconnectWiFi(); 
  handleSerialTest(); 
 
  if (millis() - lastRead >= 200) 
  { 
    lastRead = millis(); 
    readSensor(); 
    checkAlarm(); 
  } 
 
  if (millis() - lastLCD >= 250) 
  { 
    lastLCD = millis(); 
    updateLCD(); 
  } 
 
  delay(10); 
} 

